

ADVANCED TIME SERIES ECONOMETRICS

LAB 1

February 23, 2022

I am happy to briefly discuss the Stata interface, then move to the questions.




Figure 1: Stata interface.

1 Question 1

1.1 local level (unmoored inflation expectations) model

The measurement equation is:

$$y_t = \alpha_t + \varepsilon_t, \quad (1)$$

where α_t is trend inflation.

The state equation is:

$$\alpha_{t+1} = \alpha_t + u_t. \quad (2)$$

Stata:

`sspace` actually estimates the local level model:

```
13 *Estimate the local level (unmoored inflation expectations) model
14
15 *constraints impose appropriate forms for system matrices-----
16
17 //FIRST WE IMPOSE A CONSTRAINT ON THE STATE EQUATION (IE THE ALPHA EQUATION)
18 //THE COEFFICIENT ON LAGGED ALPHA IS ONE
19
20 //SECOND WE IMPOSE A CONSTRAINT ON THE MEASUREMENT EQUATION (IE THE CPIINFL EQUATION)
21 //THE COEFFICIENT OF ALPHA IS ONE
22
23 //SEE BOTTOM OF SLIDE 48 FOR RELATIONSHIP WITH GENERAL NORMAL LINEAR STATE SPACE MODEL
24
25 constraint 1 [alpha]L.alpha = 1
26 constraint 2 [CPIINFL]alpha = 1
27
28 *sspace actually estimates the local level model-----
29
30 //SPECIFY WHAT'S IN THE STATE EQUATION FIRST
31 //THEN SPECIFY WHAT'S IN THE MEASUREMENT (AKA OBSERVATION) EQUATION
32 //THEN IMPOSE THE CONSTRAINTS
33
34 sspace (alpha L.alpha, state noconstant) (CPIINFL alpha, noconstant), constraints(1 2)
35
36 //DISPLAY INFORMATION CRITERIA REQUIRED TO COMPARE MODELS, LOWER IS BETTER
37
38 estat ic
```

Figure 2: Estimate the local level model in Stata.

1.2 The anchored model

The measurement equation is:

$$y_t = \alpha_t + \varepsilon_t, \quad (3)$$

where α_t is trend inflation.

The state equation is:

$$\alpha_{t+1} = c + \rho\alpha_t + u_t. \quad (4)$$

Stata:

We can use **sspace** to estimate:

```
68 *now do a variant of the anchored inflation expectations model-----
69
70 //NOW WE DON'T CONSTRAIN THE COEFFICIENT ASSOCIATED WITH LAGGED ALPHA IN THE
71 //STATE SPACE EQUATION TO BE ONE AND WE HAVE A CONSTANT IN THE STATE EQUATION
72
73 constraint 3 [CPIINFL]alpha1 = 1
74 sspace (alpha1 L.alpha1, state) (CPIINFL alpha1, noconstant), constraints(3)
```

Figure 3: Estimate the anchored model in Stata.

1.3 Filtered and Smoothed Estimates

Filtered Estimates = estimate a state at time t using data up to time t . This is good for real time forecasting, i.e. make forecast for 1979 using data up to 1979.

Smoothed Estimates = estimate a state at time t using data up to time T (i.e. the full sample). This is good for understanding and analysing economic crises which have already happened.

Stata:

smethod(smooth) causes predict to estimate the states at each time period using all the sample data by the Kalman smoother.

smethod(filter) causes predict to estimate the states at each time period using previous and contemporaneous data by the Kalman filter. The Kalman filter is performed on previous periods and the current period.

```
40 *The following line creates smoothed estimates of the states-----
41
42 ///smethod(smooth) causes predict to estimate the states at each time period
43 ///using all the sample data by the Kalman smoother.
44 predict alpha_smooth, states smethod(smooth) equation(alpha) rmse(rmse)
45 scalar z = invnormal(.95)
46 gen lb = alpha_smooth - z*(rmse)
47 gen ub = alpha_smooth + z*(rmse)
48 tsline alpha_smooth lb ub, xtitle("") title("Smoothed Alpha Estimates with 95% Confidence Interval - Local-Level")
49 graph export Figure2.png
50
51
52 *The following creates filtered estimates of the states
53
54 ///smethod(filter) causes predict to estimate the states at each time period using
55 ///previous and contemporaneous data by the Kalman filter. The Kalman filter is
56 ///performed on previous periods and the current period.
57 predict alpha_filt, states smethod(filter) equation(alpha) rmse(rmse1)
58 gen lb1 = alpha_filt - z*(rmse1)
59 gen ub1 = alpha_filt + z*(rmse1)
60 tsline alpha_filt lb1 ub1, xtitle("") title("Filtered Alpha Estimates with 95% Confidence Interval")
61 graph export Figure3.png
```

Figure 4: Filtered and Smoothed Estimates for Local Level model in Stata.

```
77 *This creates smoothed estimates of the states
78 predict alpha_smooth1, states smethod(smooth) equation(alpha1)
79
80 *This creates filtered estimates of the states
81 predict alpha_filt1, states smethod(filter) equation(alpha1)
82 tsline CPIINFL alpha_smooth1 alpha_filt1, xtitle("") title("Anchored Inflation Expectations")
```

Figure 5: Filtered and Smoothed Estimates for the anchored model in Stata.

2 Question 2

2.1 AR(1) model in state space form

The AR (1) model in deviations from mean form can be written as a state space model:

$$y_t = \mu + u_t, \quad (5)$$

$$u_t = \rho u_{t-1} + \varepsilon_t, \quad (6)$$

where the first equation is the measurement equation and the second equation is a state equation involving the state u_t .

Stata: THE KEY PART OF THIS QUESTION IS DISCOVERING THE NOERROR OPTION

```
93 constraint 4 [CPIINFL]u = 1
94 sspace (u L.u, state noconstant) (CPIINFL u, noerror), constraints(4)
95 estat ic
96 predict yhat
97 mean yhat
```

Figure 6: AR(1) model in state space form in Stata.

2.2 conventional AR(1) model

The conventional AR(1) model:

$$y_t = \alpha + \rho y_{t-1} + \varepsilon_t. \quad (7)$$

Stata: **arima** can estimate it

```
99 arima CPIINFL, ar(1)
100 estat ic
101 predict yhat2
102 mean yhat2
```

Figure 7: Conventional AR(1) model in Stata.

3 Question 3

3.1 Local Linear Trend model

$$y_t = \alpha_t + \varepsilon_t, \quad (8)$$

$$\alpha_{t+1} = \alpha_t + \beta_t + u_{1t}, \quad (9)$$

$$\beta_{t+1} = \beta_t + u_{2t}, \quad (10)$$

where we have an additional equation (10).

Stata: `sspace` can estimate the local linear trend model, and get the smoothed estimates and filtered estimates:

```
104 //QUESTION 3i
105
106 constraint 5 [beta]_L.beta = 1
107 constraint 6 [alpha2]_L.alpha2 = 1
108 constraint 7 [alpha2]_L.beta = 1
109 constraint 8 [CPIINFL]alpha2 = 1
110
111 sspace (beta L.beta, state noconstant) (alpha2 L.alpha2 L.beta, state noconstant) (CPIINFL alpha2, noconstant), constraints (5 6 7 8) iter(100)
112
113 // //QUESTION 3iii
114 estat ic
115 //
116 // //QUESTION 3ii
117 predict alpha_smooth2, states smethod(smooth) equation(alpha2)
118 predict alpha_filt2, states smethod(filter) equation(alpha2)
119 tsline CPIINFL alpha_smooth2 alpha_filt2, xtitle("") title("Anchored Inflation Expectations")
```

Figure 8: Local Linear Trend model in Stata.

3.2 Local Linear Trend model

$$y_t = \alpha_t + \varepsilon_t, \quad (11)$$

$$\alpha_{t+1} = \alpha_t + \beta_t + u_{1t}, \quad (12)$$

$$\beta_{t+1} = \beta_t + u_{2t}, \quad \text{var}(u_{2t}) = 0, \quad (13)$$

where we have an error term with zero variance in equation (13). This means that β is constant. Again we can use the `noerror` option in **Stata**:

```

123 //QUESTION 3iv
124
125 //ERROR TERM WITH ZERO VARIANCE = CONSTANT SO AGAIN WE MAKE USE OF THE NOERROR OPTION
126
127 constraint 9 [beta1]_L.beta1 = 1
128 constraint 10 [alpha3]_L.alpha3 = 1
129 constraint 11 [alpha3]_L.beta1 = 1
130 constraint 12 [CPIINFL]alpha3 = 1
131
132 sspace (beta1 L.beta1, state noerror) (alpha3 L.alpha3 L.beta1, state noconstant) (CPIINFL alpha3, noconstant), constraints (9 10 11 12)
133 estat ic
134
135 predict alpha_smooth3, states smethod(smooth) equation(alpha3) rmse(rmse2)
136 predict alpha_filt3, states smethod(filter) equation(alpha3)
137 tsline CPIINFL alpha_smooth3 alpha_filt3, xtitle("") title("Anchored Inflation Expectations")
138 graph export Figure5.png
139
140 gen lb2 = alpha_smooth3 - z*(rmse2)
141 gen ub2 = alpha_smooth3 + z*(rmse2)
142 tsline alpha_smooth3 lb2 ub2, xtitle("") title("Smoothed Alpha Estimates with 95% Confidence Interval (local-Linear Trend Model)")
143 graph export Figure6.png

```

Figure 9: Restricted Local Linear Trend model in Stata.